Efficient Portfolios in Excel Using the Solver and Matrix Algebra

This note outlines how to use the solver and matrix algebra in Excel to compute efficient portfolios. The example used in this note is in the spreadsheet 3firmExample.xlsx, and is the same example used in the lecture notes titled “Portfolio Theory with Matrix Algebra”.

Last updated: November 24, 2009

The Solver Add-In

The solver is an Excel Add-In created by Frontline Systems (www.solver.com) that can be used to solve general optimization problems that may be subject to certain kinds of constraints. In this note we show how it can be used to find portfolios that minimize risk subject to certain constraints.

The solver add-in must be activated before it can be used within Excel. In Excel 2007, you activate add-ins by clicking on the office button and then clicking on the Excel Options box at the bottom of the menu.



This opens the Excel options dialogue box. Click Add-Ins, which displays the available Add-Ins for Excel. Make sure the Solver Add-In is an Active Application Add-In.



Matrix Algebra in Excel

Excel has several built-in array formulas that can perform basic matrix algebra operations. The main functions are listed in table below

Array Function	Description
MINVERSE	Compute inverse of matrix
MMULT	Matrix multiplication
TRANSPOSE	Compute transpose of matrix

To evaluate an array function in Excel, you must use the magic key stoke combination: <CTRL>-<SHIFT>-<ENTER> (hold down all three keys at once then release).

Example Data

In the Data tab of the spreadsheet 3firmExample.xls is the example monthly return data on three assets: Microsoft, Nordstrom and Starbucks. The monthly means and covariance matrix of the returns are computed and these are referenced as the input data on the portfolio tab as illustrated in the screen shot below.



	A	B	C	D	E	F	G	H	I
1									
2	asset	muvec	sdvals	sigma	microsoft	nordstrom	starbucks		rf
3	microsoft	4.27%	10.0%	microsoft	0.0100	0.0018	0.0011		0.005
4	nordstrom	0.15%	10.5%	nordstrom	0.0018	0.0109	0.0026		
5	starbucks	2.85%	14.1%	starbucks	0.0011	0.0026	0.0199		

In the spreadsheet, cells colored light blue contain input data (fixed data not created by some formula) and cells colored tan contain output data (data created by applying some formula). Also, some cells are explicitly named. For example the range of cells B3:B5 is named muvec. If these cells are highlighted then muvec will appear in the Name Box in the upper left hand corner of the spreadsheet. Similarly, the range of cells E3:G5 is named sigma. For matrix algebra calculations, it is convenient to use named ranges in array formulas.

The Global Minimum Variance Portfolio

The global minimum variance portfolio solves the optimization problem

$$\min_{\mathbf{m}} \ \sigma_{p,m}^2 = \mathbf{m}'\Sigma\mathbf{m} \text{ s.t. } \mathbf{m}'\mathbf{1} = 1$$

This optimization problem can be solved easily using the solver with matrix algebra functions. The screen shot of the portfolio tab below shows how to set-up this optimization problem in Excel.



The range of cells D10:D12 is called mvec and will contain the weights in the minimum variance portfolio once the solver is run and the solution to the optimization problem is found. Before the solver is to be run, these cells should contain an initial guess of the minimum variance portfolio. A simple guess for this vector whose weights sum to one is $m_{msft} = 0.3, m_{nord} = 0.3, m_{sbux} = 0.4.$

To use the solver, a cell containing the function to be maximized or minimized must be specified. Here, this cell is F10 which contains the array formula

{=MMULT(TRANSPOSE(mvec),MMULT(sigma,mvec))}

which evaluates the matrix algebra formula for the variance of a portfolio: $\sigma^2_{p,m} = \mathbf{m}'\Sigma\mathbf{m}$. Notice that the formula is surrounded by curly braces {}. This indicates that <CTRL>-<Shift>-<Enter> was used to evaluate the formula so that it is to be interpreted as an array formula. If you don't see the curly braces then the formula will not be evaluate correctly. We also need a cell to contain a formula that will be used to impose the constraint that the portfolio weights sum to one: $\mathbf{m}'\mathbf{1} = m_{msft} + m_{nord} + m_{sbux} = 1.$ This formula is specified in cell E10 as

=SUM(mvec)

The solver add-in is located on the data tab of the top menu ribbon in the right hand corner. To run the solver, click the cell containing the formula you want to optimize (cell F10, and named sig2px) and then click on the solver button. This will open up the solver dialogue box as shown below.



The field named Set Target Cell must contain either the name or the reference to the cell containing the formula to optimize. You have three choices for the type of optimization: Max, Min and Value of. Here, we want to minimize the portfolio variance so Min should be selected. Next, we must specify the cells containing the variables which are being optimized. These are specified in the By Changing Cells field. Here, we can type in the name mvec or specify the range of cells D10:D12. Finally, we must Add the constraint that the weights sum to one. We do this by clicking the Add button, which opens the Add Constraint dialogue box show below.



The Cell Reference contains the cell (E10) that has the formula for the constraint $\mathbf{m}'\mathbf{1} = m_{msft} + m_{nord} + m_{sbux} = 1$. We specify the value of the constraint, 1, in the Constraint field. Once everything is filled in, click OK to go back to the solver dialogue. The complete dialogue should look like one shown below.



To run the solver, click the Solve button. The computation is generally very fast. If successful, you should see the following dialogue box



The message "Solver found a solution. All constraints and optimality conditions are satisfied" means that the first and second order conditions for a minimum are satisfied. Click the Keep Solver Solution option button and then click OK. Your spreadsheet should look like the one below.



The global minimum variance portfolio has 44% in Microsoft, 36% in Nordstrom and 19% in Starbucks. The expected return on this portfolio is given in cell C13 (called mupx) and is computed using the formula $\mu_{p,m} = \mathbf{m}'\mu$. The Excel array formula is

{=MMULT(TRANSPOSE(mvec),muvec)}

The portfolio standard deviation in cell C14 is the square root of the portfolio variance, sig2px, in cell F10.

Minimum Variance Portfolio subject to Target Expected Return

A minimum variance portfolio with target expected return equal to μ_0 solves the optimization problem

$$\min_{y} \ \sigma^2_{p,y} = \mathbf{y}'\Sigma\mathbf{y} \text{ s.t. } \mathbf{y}'\mu = \mu_0 \text{ and } \mathbf{y}'\mathbf{1} = 1$$

This optimization problem can also be easily solved using the solver with matrix algebra functions. The screenshot below shows how to set-up this optimization problem in Excel where the target expected return is the expected return on Microsoft (4.27%).



The range of cells K10:K12 is called yvec and will contain the weights in the efficient portfolio once the solver is run and the solution to the optimization problem is found. Before the solver is to be run, these cells should contain an initial guess of the minimum variance portfolio. A simple guess for this vector whose weights sum to one is $y_{msft} = 0.3, y_{nord} = 0.3, y_{sbux} = 0.4$. The cell containing the formula for portfolio variance, $\sigma^2_{p,y} = \mathbf{y}'\Sigma\mathbf{y}$, is in cell O10 which contains the array formula

{=MMULT(TRANSPOSE(yvec),MMULT(sigma,yvec))}

We also need two additional cells to contain formulas that will be used to impose the constraints that the portfolio expected return is equal to the target return, $\mu_{p,y} = \mathbf{y}'\mu = \mu_0$, and that the portfolio weights sum to one, $\mathbf{y}'\mathbf{1} = y_{msft} + y_{nord} + y_{sbux} = 1$. These formulas are specified in cells L10 and N10, which contain the Excel formulas =SUM(yvec) and {=MMULT(TRANSPOSE(yvec),muvec)}, respectively.

To run the solver, click cell O10 (called sig2py) and then click on the solver button. Make sure the solver dialogue box is filled out to look like the one below.



Notice that there are now two constraints specified. The first one imposes $\mathbf{y}'\mathbf{1} = y_{msft} + y_{nord} + y_{sbux} = 1$, and the second one imposes $\mu_{p,y} = \mathbf{y}'\mu = \mu_0 = \mu_{msft} = 0.0475$. To run the solver, click the Solve button. You should see a dialogue box that says that the solver found a solution and that all optimality conditions are satisfied. Keep the solution and click OK. Your spreadsheet should look like the one below.



The efficient portfolio has weights $y_{msft} = 0.83, y_{nord} = -0.09, y_{sbux} = 0.26.$ Notice that Nordstrom is sold short in this portfolio because it has a negative weight. The expected return on this portfolio is equal to the target expected return (see cell N10 named mupy) and the weights sum to one. Notice that the standard deviation of this portfolio (see cell P10) is smaller than the standard deviation of Microsoft (see cell C3).

Computing the Efficient Frontier of Risky Assets

The efficient frontier of risky assets can be constructed from any two efficient portfolios. A natural question to ask is which two efficient portfolios should be used? I find that the following two efficient portfolios leads to the easy creation of the efficient frontier:

1. Efficient portfolio 1: global minimum variance portfolio
2. Efficient portfolio 2: efficient portfolio with target expected return equal to the highest average return among the assets under consideration.

For the current example, the asset with the highest average return is Microsoft (average return is 4.27%) and we already computed the efficient portfolio with target expected return equal to the average return on Microsoft.

Given any two efficient portfolios with weight vectors **m** and **y** the convex combination

$$\mathbf{z} = \alpha \cdot \mathbf{m} + (1-\alpha) \cdot \mathbf{y}$$

for any constant α is also an efficient portfolio. The expected return and variance of this portfolio are

$$\mu_{p,z} = \alpha \cdot \mu_{p,m} + (1-\alpha) \cdot \mu_{p,y}$$
$$\sigma_{p,z}^2 = \alpha^2 \sigma_{p,m}^2 + (1-\alpha)^2 \sigma_{p,y}^2 + 2\alpha(1-\alpha)\sigma_{my},$$

where the covariance between the returns on portfolios **m** and **y** is computed using $\sigma_{my} = \mathbf{m}'\Sigma\mathbf{y}$. To create the efficient frontier, create a grid of α values starting at 1 and decrease in increments of 0.1. Use as many values in the grid as necessary to make a nice plot.

A screenshot of the part of the spreadsheet to create these portfolios is shown below.



3firmExample - Microsoft Excel

P20 =N20*mupx+O20*mupy

$z = \alpha \cdot m + (1-\alpha) \cdot y$ → Frontier Portfolios

	N	O	P	Q	R	S	T	U
18						microsoft	nordstrom	starbucks
19	alpha	1-alpha	E[Rp,z]	var(Rp,z)	SD(Rp,z)	z_msft	z_nord	z_sbux
20	1	0	2.50%	0.00528	7.27%	44%	36%	19%
21	0.9	0.1	2.68%	0.00531	7.29%	48%	32%	20%
22	0.8	0.2	2.85%	0.00541	7.35%	52%	27%	21%
23	0.7	0.3	3.03%	0.00556	7.46%	56%	23%	21%
24	0.6	0.4	3.21%	0.00578	7.60%	60%	18%	22%
25	0.5	0.5	3.38%	0.00605	7.78%	64%	14%	23%
26	0.4	0.6	3.56%	0.0064	8.00%	67%	9%	23%
27	0.3	0.7	3.74%	0.0068	8.24%	71%	5%	24%
28	0.2	0.8	3.92%	0.00726	8.52%	75%	0%	25%
29	0.1	0.9	4.09%	0.00779	8.82%	79%	-5%	26%
30	0	1	4.27%	0.00837	9.15%	83%	-9%	26%
31	-0.1	1.1	4.45%	0.00902	9.50%	87%	-14%	27%

Consider the first convex combination with $\alpha = 1$. This portfolio is the global minimum variance portfolio. The cell P20 contains the formula =N20*mupx+O20*mupy for the expected portfolio return, and the cell Q20 contains the formula =N20^2*sig2px+O20^2*sig2py+2*N20*O20*sigmaxy for the portfolio variance. The covariance term sigmaxy is computed in the cell R9 (not shown) which contains the array formula {=MMULT(MMULT(TRANSPOSE(mvec),sigma),yvec)}. The cells S20:U20 give the weights in the convex combination computed using the array formula {=TRANSPOSE(N20*D10:D12+O20*K10:K12)}.

The efficient frontier can be plotted by making a scatter plot with the expected return values (cells P20:P50) on the y-axis and the standard deviation values (cells R20:R50) on the horizontal axis.

Computing the Tangency Portfolio

The tangency portfolio is the portfolio of risky assets that has the highest Sharpe's slope. This portfolio can be found by solving the optimization problem

$$\max_{t} \frac{\mathbf{t}'\mu - r_f}{(\mathbf{t}'\Sigma\mathbf{t})^{1/2}} \text{ s.t. } \mathbf{t}'\mathbf{1} = 1$$

This optimization problem can also be easily solved using the solver with matrix algebra functions. The screenshot below shows how to set-up this optimization problem in Excel.



The range of cells D33:D35 is called tvec and will contain the weights in the tangency portfolio once the solver is run and the solution to the optimization problem is found. Before the solver is to be run, these cells should contain an initial guess of the minimum variance portfolio. A simple guess for this vector whose weights sum to one is $t_{msft} = 0.3, t_{nord} = 0.3, t_{sbux} = 0.4$. The computation of Sharpe's slope is broken down into two pieces. The first piece is the numerator of Sharpe's slope, $\mu_{p,t} - r_f = \mathbf{t}'\mu - r_f$, and is computed in cell F33 using the array formula {=MMULT(TRANSPOSE(tvec),muvec)-rf}. The second piece is the square of the denominator of Sharpe's slope, $\sigma^2_{p,t} = \mathbf{t}'\Sigma\mathbf{t}$, and is computed in cell G33 using the array formula {=MMULT(TRANSPOSE(tvec),MMULT(sigma,tvec))}. Finally, Sharpe's slope is evaluated in cell H33 using the formula =F33/SQRT(G33). This is the cell that is passed to the solver.

To run the solver, click cell H33 and then click on the solver button. Make sure the solver dialogue box is filled out to look like the one below.



Make sure that the Max button is selected because we want to maximize the Sharpe's slope. To run the solver, click the Solve button. You should see a dialogue box that says that the solver found a solution and that all optimality conditions are satisfied. Keep the solution and click OK. Your spreadsheet should look like the one below.



The tangency portfolio has weights $t_{msft} = 1.03, t_{nord} = -0.32, t_{sbux} = 0.30$. Notice that Nordstrom is sold short in this portfolio because it has a negative weight. The expected return on this portfolio, $\mu_{p,t} = \mathbf{t}'\mu$, is given in cell C36 (called mut) and is computed using the array formula {=MMULT(TRANSPOSE(tvec),muvec)}.

Computing Efficient Portfolios of T-Bills and Risky Assets

From the mutual fund separation theorem, the efficient portfolios of T-Bills and risky assets are combinations of T-Bills and the tangency portfolio. The expected return and standard deviation values of these portfolios are computed using

$$\mu_p^e = r_f + x_{\tan}\left(\mu_{\tan} - r_f\right)$$
$$\sigma_p^e = x_{\tan}\sigma_{\tan}$$

A screenshot of the spreadsheet where these portfolios are computed is given below.



The portfolio with $x_{\tan} = 0$ is shown in the cells J34:L34. The expected return is computed in cell K34 and is given by the formula =rf+J34*(mut-rf). The standard deviation is computed in cell L34 and is given by the formula =J34*sigt. The named range sigt is the standard deviation of the tangency portfolio and is given in cell C37.

Efficient Portfolios with No Short Sales Constraints

In many situations short sales of assets are not allowed. Recall, a short sale of an asset occurs when you borrow the asset and then sell it. The proceeds of the short sale are usually used to finance the purchase of other assets. Because the asset was borrowed it eventually has to be returned. You do this by repurchasing the asset at some time in the future and then returning the asset to whomever you borrowed it from. You make a profit on a short sale if the price of the asset drops during the period of time you have borrowed the asset because you repurchase the asset for a price less than for what you originally sold it. In the context of portfolio theory, when you short sell an asset the corresponding portfolio weight is negative. Hence, when short sales are prohibited all of the portfolio weights must be constrained to be positive. This type of non-negativity constraint is easy to impose in the solver.

Minimum Variance Portfolio subject to Target Expected Return with No Short Sales

A minimum variance portfolio with target expected return equal to μ_0 and no short sales solves the optimization problem

$$\min_{y} \ \sigma_{p,y}^2 = \mathbf{y}'\Sigma\mathbf{y} \text{ s.t. } \mathbf{y}'\mu = \mu_0, \mathbf{y}'\mathbf{1} = 1 \text{ and } y_i \geq 0$$

This optimization problem can also be easily solved using the solver with matrix algebra functions. The screenshot below shows how to set-up this optimization problem in Excel where the target expected return is the expected return on Microsoft (4.27%). Previously, we solved this problem where we allowed for short sales. In that case, the efficient portfolio was $y_{msft} = 0.83, y_{nord} = -0.09, y_{sbux} = 0.26.$

Notice that Nordstrom was sold short in that portfolio. Now we want to impose the no short sales restrictions. We set up the Excel spreadsheet exactly how we did before. The only difference occurs in how we use the solver. We add an additional constraint that forces all of the portfolio weights to be positive. The screen shot below shows the initial set-up.



The range of cells AC5:AC7 is called wvec and will contain the weights in the efficient portfolio once the solver is run and the solution to the optimization problem is found. Before the solver is to be run, these cells should contain an initial guess of the minimum variance portfolio. The cell containing the formula for portfolio variance, $\sigma_{p,w}^2 = \mathbf{w}'\Sigma\mathbf{y}$, is in cell AG5 which contains the array formula

{=MMULT(TRANSPOSE(wvec),MMULT(sigma,wvec))}

The cells AD5 and AF5 contain the Excel formulas =SUM(wvec) and {=MMULT(TRANSPOSE(wvec), muvec)} that will be used to impose the restrictions that the portfolio weights sum to one and that the portfolio expected return is equal to the target return. To run the solver, click cell AG5 and then click on the solver button. Make sure the solver dialogue box is filled out to look like the one below.



To add the no short sales constraints, click the Add button to open the Add Constraint dialogue. The no short sales constraints inequality constraints on the elements of wvec:



Click OK. The final solver dialogue should look like the one below.



You should see a dialogue box that says that the solver found a solution and that all optimality conditions are satisfied. Keep the solution and click OK. Your spreadsheet should look like the one below.



The efficient portfolio has 100% in Microsoft and 0% in the other assets.